You can download the .nb file here.

```
ClearAll[X, Y, Z, h, i, j, k, l, m, n]


(* Auxiliary definitions *)
biased[x_, y_, h_, i_, j_] :=
 Probability[X == h ⋀ Y == i ⋀ Z == j, {X ≈ BernoulliDistribution[y],
   Y ≈ BernoulliDistribution[y], Z ≈ BernoulliDistribution[y]}]
fair[x_, y_, h_, i_, j_] := Probability[X == h ⋀ Y == i ⋀ Z == j,
  {X ≈ BernoulliDistribution[.5],
   Y ≈ BernoulliDistribution[.5], Z ≈ BernoulliDistribution[.5]}]
prob1[x_, y_, h_, i_, j_, k_] := If[k == 0 ∨ y == 0.5, 0, biased[x, y, h, i, j]]
prob2[x_, y_, h_, i_, j_, k_] := If[k == 1, 0, fair[x, y, h, i, j]]
Tt[h_, i_, j_, k_, x_, y_, z_, w_] := If[h == x ⋀ i == y ⋀ j == z ⋀ k == w, 1, 0]


(* A possible world here is a tuple <h,i,j,k>, where h is the truth value of H1,
i the truth value of H2, j the truth value of H3, and k the truth value of B.*)

(* cred[x,y,h,i,j,k] is the credence in world <h,i,j,k>
 given that your credence that the coin has a bias of y towards heads is x*)
cred[x_, y_, h_, i_, j_, k_] := If[y == 0.5, prob2[x, y, h, i, j, k],
  x * prob1[x, y, h, i, j, k] + (1 - x) * prob2[x, y, h, i, j, k]]

(* credH1 (resp. credT1) is the result of conditionalizing on H1 (resp. T1). *)
credH1[x_, y_, h_, i_, j_, k_] := If[h == 1, cred[x, y, h, i, j, k] /
    (Sum[cred[x, y, h, l, m, n], {l, 0, 1}, {m, 0, 1}, {n, 0, 1}]), 0]
credT1[x_, y_, h_, i_, j_, k_] := If[h == 0, cred[x, y, h, i, j, k] /
    (Sum[cred[x, y, h, l, m, n], {l, 0, 1}, {m, 0, 1}, {n, 0, 1}]), 0]

(* altBrier[P,x,y,h,i,j,k]  (resp. altLog[P,x,y,h,i,j,k]) is the Brier score
  (resp. Log score) of probability function P at world <h,i,j,k> *)
(* altEEU[S,X,Y,x,y] is the expected S-
 value of probability function X relative to probability function Y.*)
altBrier[func_Symbol, x_, y_, h_, i_, j_, k_] :=
 -Sum[(func[x, y, l, m, n, o] - Tt[l, m, n, o, h, i, j, k])^2,
   {l, 0, 1}, {m, 0, 1}, {n, 0, 1}, {o, 0, 1}]
altLog[func_Symbol, x_, y_, h_, i_, j_, k_] :=
 If[func[x, y, h, i, j, k] == 0, 0, Log[func[x, y, h, i, j, k]]]
altEEU[S_, X_, Y_, x_, y_] :=
 Sum[If[Y[x, y, h, i, j, k] == 0, 0, Y[x, y, h, i, j, k] * S[X, x, y, h, i, j, k]],
  {h, 0, 1}, {i, 0, 1}, {j, 0, 1}, {k, 0, 1}]


heads1[x_, y_] := Sum[cred[x, y, 1, i, j, k], {i, 0, 1}, {j, 0, 1}, {k, 0, 1}]
isFair[x_, y_] := Sum[cred[x, y, h, i, j, 0], {h, 0, 1}, {i, 0, 1}, {j, 0, 1}]
isBiased[x_, y_] := Sum[cred[x, y, h, i, j, 1], {h, 0, 1}, {i, 0, 1}, {j, 0, 1}]


(* Value of the questions ?B and ?H1 using
 the Brier score as epistemic utility function.*)
altValBiasB[x_, y_] :=
 x * (altEEU[altBrier, prob1, prob1, x, y] - altEEU[altBrier, cred, prob1, x, y]) +
  (1 - x) * (altEEU[altBrier, prob2, prob2, x, y] - altEEU[altBrier, cred, prob2, x, y])
altValHeadsB[x_, y_] := heads1[x, y] * (altEEU[altBrier, credH1, credH1, x, y] -
```

```
      altEEU[altBrier, cred, credH1, x, y]) + (1 - heads1[x, y]) *
    (altEEU[altBrier, credT1, credT1, x, y] - altEEU[altBrier, cred, credT1, x, y])

(* Value of the questions ?B and ?H1 using
 the log score as epistemic utility function.*)
altValBiasL[x_, y_] :=
 x * (altEEU[altLog, prob1, prob1, x, y] - altEEU[altLog, cred, prob1, x, y]) +
  (1 - x) * (altEEU[altLog, prob2, prob2, x, y] - altEEU[altLog, cred, prob2, x, y])
altValHeadsL[x_, y_] := heads1[x, y] * (altEEU[altLog, credH1, credH1, x, y] -
      altEEU[altLog, cred, credH1, x, y]) + (1 - heads1[x, y]) *
   (altEEU[altLog, credT1, credT1, x, y] - altEEU[altLog, cred, credT1, x, y])

altAskBiasB[x_, y_] := If[altValBiasB[x, y] ≥ altValHeadsB[x, y], 1, 0]
altAskBiasL[x_, y_] := If[altValBiasL[x, y] ≥ altValHeadsL[x, y], 1, 0]


(*Value of each of ?B and ?H1 as a function of your credence in B,
using the Brier score.*)
Table[Round[altValBiasB[x, 0.2], 0.001], {x, 0.1, 0.9, 0.1}]
Table[Round[altValHeadsB[x, 0.2], 0.001], {x, 0.1, 0.9, 0.1}]


{0.04, 0.07, 0.092, 0.105, 0.11, 0.105, 0.092, 0.07, 0.04}

{0.105, 0.093, 0.086, 0.084, 0.086, 0.09, 0.097, 0.108, 0.124}

(*Value of each of ?B and ?H1 as a function of your credence in B,
using the log score.*)
Table[Round[altValBiasL[x, 0.2], 0.001], {x, 0.1, 0.9, 0.1}]
Table[Round[altValHeadsL[x, 0.2], 0.001], {x, 0.1, 0.9, 0.1}]


{0.325, 0.5, 0.611, 0.673, 0.693, 0.673, 0.611, 0.5, 0.325}

{0.691, 0.686, 0.677, 0.664, 0.647, 0.627, 0.602, 0.573, 0.539}

(* Determine whether the value of ?B is at least as high as that of ?H1,
when the bias is y towards heads,
using the Brier score and using the log score. *)
Agree[x_, y_] := If[altAskBiasB[x, y] == altAskBiasL[x, y], 1, 0]
rowAgree[y_] := Prepend[Table[Agree[x, y], {x, 0.1, 0.9, 0.1}], y]

Table[rowAgree[y], {y, 0.1, 0.9, 0.1}]


{{0.1, 1, 0, 0, 1, 1, 1, 1, 1, 1},
 {0.2, 1, 1, 0, 1, 1, 1, 0, 1, 1}, {0.3, 1, 1, 1, 0, 1, 1, 1, 1, 1},
 {0.4, 1, 1, 1, 1, 1, 1, 1, 1, 1}, {0.5, 1, 1, 1, 1, 1, 1, 1, 1, 1},
 {0.6, 1, 1, 1, 1, 1, 1, 1, 1, 1}, {0.7, 1, 1, 1, 0, 1, 1, 1, 1, 1},
 {0.8, 1, 1, 0, 1, 1, 1, 0, 1, 1}, {0.9, 1, 0, 0, 1, 1, 1, 1, 1, 1}}
```

TableForm[%111]

0.1	1	0	0	1	1	1	1	1	1
0.2	1	1	0	1	1	1	0	1	1
0.3	1	1	1	0	1	1	1	1	1
0.4	1	1	1	1	1	1	1	1	1
0.5	1	1	1	1	1	1	1	1	1
0.6	1	1	1	1	1	1	1	1	1
0.7	1	1	1	0	1	1	1	1	1
0.8	1	1	0	1	1	1	0	1	1
0.9	1	0	0	1	1	1	1	1	1